Здравствуйте, форумчане.
Прошу не кидать в меня тапки за мою серость. В последнее время занимаюсь скорее управлением, чем программированием, поэтому стала испытывать сложности с последним.
Конфигурация Управление производственным предприятием, редакция 1.3 (1.3.143.1)
Задача: выдавать спецодежду согласно перечня полагающейся сп/о и контролировать сроки эксплуатации. А именно, менеджер, занимающийся снабжением, должен иметь отчет, в котором указана информация о сотруднике (включая размеры одежды и обуви), о полагающемся наборе сп/о, о фактически выданной сп/о, дате выдачи и ориентировочной дате следующей выдачи согласно срока эксплуатации)
Для реализации данного функционала использован типовой документ "Передача материалов в эксплуатацию", а также создан справочник "пакет спецодежды" и регистр сведений "Потребность в спецодежде" (содержит размеры и пакет одежды для каждого сотрудника, которому положена сп/о).
За основу отчета взяла запрос из отчета "Ведомость по материалам в эксплуатации", попыталась его допилить.

```1c
    ВЫБРАТЬ
    РаботникиОрганизации.Сотрудник,
    РаботникиОрганизации.Сотрудник.Физлицо,
    РаботникиОрганизации.Сотрудник.ТекущаяДолжностьКомпании,
    ВЫБОР
    КОГДА ВЫБОР
    КОГДА РаботникиОрганизации.ПериодЗавершения <= &Период
    И РаботникиОрганизации.ПериодЗавершения <> ДАТАВРЕМЯ(1, 1, 1, 0, 0, 0)
    ТОГДА РаботникиОрганизации.ПричинаИзмененияСостоянияЗавершения
    ИНАЧЕ РаботникиОрганизации.ПричинаИзмененияСостояния
    КОНЕЦ = ЗНАЧЕНИЕ(Перечисление.ПричиныИзмененияСостояния.Увольнение)
    ТОГДА ЗНАЧЕНИЕ(Перечисление.СостоянияРаботникаОрганизации.НеРаботает)
    ИНАЧЕ ЕСТЬNULL(ВЫБОР
    КОГДА &Период >= СостояниеРаботниковОрганизации.ПериодЗавершения
    И СостояниеРаботниковОрганизации.ПериодЗавершения <> ДАТАВРЕМЯ(1, 1, 1, 0, 0, 0)
    ТОГДА СостояниеРаботниковОрганизации.СостояниеЗавершения
    ИНАЧЕ СостояниеРаботниковОрганизации.Состояние
    КОНЕЦ, ЗНАЧЕНИЕ(Перечисление.СостоянияРаботникаОрганизации.Работает))
    КОНЕЦ КАК Состояние,
    РаботникиОрганизации.Сотрудник.ТекущееПодразделениеКомпании,
    РаботникиОрганизации.Должность
    ПОМЕСТИТЬ СписокРаботников
    ИЗ
    РегистрСведений.РаботникиОрганизаций.СрезПоследних КАК РаботникиОрганизации
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.СостояниеРаботниковОрганизаций.СрезПоследних КАК СостояниеРаботниковОрганизации
    ПО РаботникиОрганизации.Организация = СостояниеРаботниковОрганизации.Организация
    И РаботникиОрганизации.Сотрудник = СостояниеРаботниковОрганизации.Сотрудник
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ПотребностьВСпецОдеждеСрезПоследних.ФизЛицо,
    ПотребностьВСпецОдеждеСрезПоследних.РазмерОбуви,
    ПотребностьВСпецОдеждеСрезПоследних.РазмерОдежды,
    ПотребностьВСпецОдеждеСрезПоследних.Рост,
    ПакетСпецодеждыСписокСпецодежды.Номенклатура,
    ПакетСпецодеждыСписокСпецодежды.НазначениеИспользования
    ПОМЕСТИТЬ ПотребностиИРазмеры
    ИЗ
    РегистрСведений.ПотребностьВСпецОдежде.СрезПоследних(&Период, ) КАК ПотребностьВСпецОдеждеСрезПоследних
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ПакетСпецодежды.СписокСпецодежды КАК ПакетСпецодеждыСписокСпецодежды
    ПО ПотребностьВСпецОдеждеСрезПоследних.ПакетОдежды = ПакетСпецодеждыСписокСпецодежды.Ссылка
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ПередачаМатериаловВЭксплуатациюМатериалы.Ссылка,
    ПередачаМатериаловВЭксплуатациюМатериалы.НомерСтроки,
    ПередачаМатериаловВЭксплуатациюМатериалы.ЕдиницаИзмерения,
    ПередачаМатериаловВЭксплуатациюМатериалы.Количество,
    ПередачаМатериаловВЭксплуатациюМатериалы.Коэффициент,
    ПередачаМатериаловВЭксплуатациюМатериалы.НазначениеИспользования,
    ПередачаМатериаловВЭксплуатациюМатериалы.Номенклатура,
    ПередачаМатериаловВЭксплуатациюМатериалы.СерияНоменклатуры,
    ПередачаМатериаловВЭксплуатациюМатериалы.ФизЛицо,
    ПередачаМатериаловВЭксплуатациюМатериалы.ХарактеристикаНоменклатуры,
    ПередачаМатериаловВЭксплуатациюМатериалы.НазначениеИспользования.СрокПолезногоИспользования,
    ПередачаМатериаловВЭксплуатациюМатериалы.Ссылка.Дата,
    ДОБАВИТЬКДАТЕ(ПередачаМатериаловВЭксплуатациюМатериалы.Ссылка.Дата, МЕСЯЦ, ПередачаМатериаловВЭксплуатациюМатериалы.НазначениеИспользования.СрокПолезногоИспользования) КАК ОриентировочнаяДатаСдачи
    ПОМЕСТИТЬ ЭксплуатацияСНазначениемИспользованияИСроком
    ИЗ
    Документ.ПередачаМатериаловВЭксплуатацию.Материалы КАК ПередачаМатериаловВЭксплуатациюМатериалы

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.Ссылка,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.НомерСтроки,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.ЕдиницаИзмерения,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.Количество,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.Коэффициент,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.НазначениеИспользования,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.Номенклатура,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.СерияНоменклатуры,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.ФизЛицо,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.ХарактеристикаНоменклатуры,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.НазначениеИспользования.СрокПолезногоИспользования,
    ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.ДатаПередачиВЭксплуатацию,
    ДОБАВИТЬКДАТЕ(ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.ДатаПередачиВЭксплуатацию, МЕСЯЦ, ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы.НазначениеИспользования.СрокПолезногоИспользования)
    ИЗ
    Документ.ВводНачальныхОстатковМатериаловВЭксплуатации.Материалы КАК ВводНачальныхОстатковМатериаловВЭксплуатацииМатериалы

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    СписаниеМатериаловИзЭксплуатацииМатериалы.Ссылка,
    СписаниеМатериаловИзЭксплуатацииМатериалы.НомерСтроки,
    СписаниеМатериаловИзЭксплуатацииМатериалы.ЕдиницаИзмерения,
    СписаниеМатериаловИзЭксплуатацииМатериалы.Количество,
    СписаниеМатериаловИзЭксплуатацииМатериалы.Коэффициент,
    СписаниеМатериаловИзЭксплуатацииМатериалы.НазначениеИспользования,
    СписаниеМатериаловИзЭксплуатацииМатериалы.Номенклатура,
    СписаниеМатериаловИзЭксплуатацииМатериалы.СерияНоменклатуры,
    СписаниеМатериаловИзЭксплуатацииМатериалы.ФизЛицо,
    СписаниеМатериаловИзЭксплуатацииМатериалы.ХарактеристикаНоменклатуры,
    СписаниеМатериаловИзЭксплуатацииМатериалы.НазначениеИспользования.СрокПолезногоИспользования,
    СписаниеМатериаловИзЭксплуатацииМатериалы.Ссылка.Дата,
    ДОБАВИТЬКДАТЕ(СписаниеМатериаловИзЭксплуатацииМатериалы.Ссылка.Дата, МЕСЯЦ, СписаниеМатериаловИзЭксплуатацииМатериалы.НазначениеИспользования.СрокПолезногоИспользования)
    ИЗ
    Документ.СписаниеМатериаловИзЭксплуатации.Материалы КАК СписаниеМатериаловИзЭксплуатацииМатериалы
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    СписокРаботников.Сотрудник,
    СписокРаботников.СотрудникТекущаяДолжностьКомпании,
    МатериалыВЭксплуатацииОстатки.Подразделение,
    МатериалыВЭксплуатацииОстатки.Номенклатура,
    МатериалыВЭксплуатацииОстатки.СерияНоменклатуры,
    МатериалыВЭксплуатацииОстатки.ХарактеристикаНоменклатуры,
    МатериалыВЭксплуатацииОстатки.КоличествоОстаток,
    СписокРаботников.СотрудникФизлицо,
    СписокРаботников.СотрудникТекущееПодразделениеКомпании,
    СписокРаботников.Должность
    ПОМЕСТИТЬ ОстаткиМатериаловВЭксплуатации
    ИЗ
    РегистрНакопления.МатериалыВЭксплуатации.Остатки КАК МатериалыВЭксплуатацииОстатки
    ЛЕВОЕ СОЕДИНЕНИЕ СписокРаботников КАК СписокРаботников
    ПО МатериалыВЭксплуатацииОстатки.ФизЛицо = СписокРаботников.СотрудникФизлицо
    ГДЕ
    СписокРаботников.Состояние <> &Состояние
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    МатериалыВЭксплуатацииОстаткиИОбороты.Номенклатура КАК Номенклатура,
    МатериалыВЭксплуатацииОстаткиИОбороты.ХарактеристикаНоменклатуры КАК ХарактеристикаНоменклатуры,
    МатериалыВЭксплуатацииОстаткиИОбороты.СерияНоменклатуры КАК СерияНоменклатуры,
    ОстаткиМатериаловВЭксплуатации.Сотрудник КАК Сотрудник,
    ОстаткиМатериаловВЭксплуатации.СотрудникТекущаяДолжностьКомпании КАК СотрудникТекущаяДолжностьКомпании,
    ОстаткиМатериаловВЭксплуатации.СотрудникФизлицо КАК СотрудникФизлицо,
    ЭксплуатацияСНазначениемИспользованияИСроком.НазначениеИспользования,
    ЭксплуатацияСНазначениемИспользованияИСроком.НазначениеИспользованияСрокПолезногоИспользования КАК СрокПолезногоИспользования,
    ЭксплуатацияСНазначениемИспользованияИСроком.Дата КАК Дата,
    ЭксплуатацияСНазначениемИспользованияИСроком.ОриентировочнаяДатаСдачи КАК ОриентировочнаяДатаСдачи,
    ОстаткиМатериаловВЭксплуатации.СотрудникТекущееПодразделениеКомпании КАК СотрудникТекущееПодразделениеКомпании,
    ОстаткиМатериаловВЭксплуатации.Должность
    ПОМЕСТИТЬ ОстаткиНазначениеИДатаСдачи
    ИЗ
    ОстаткиМатериаловВЭксплуатации КАК ОстаткиМатериаловВЭксплуатации
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.МатериалыВЭксплуатации.ОстаткиИОбороты(, &Период, Запись, , ) КАК МатериалыВЭксплуатацииОстаткиИОбороты
    ЛЕВОЕ СОЕДИНЕНИЕ ЭксплуатацияСНазначениемИспользованияИСроком КАК ЭксплуатацияСНазначениемИспользованияИСроком
    ПО МатериалыВЭксплуатацииОстаткиИОбороты.Регистратор = ЭксплуатацияСНазначениемИспользованияИСроком.Ссылка
    И МатериалыВЭксплуатацииОстаткиИОбороты.НомерСтроки = ЭксплуатацияСНазначениемИспользованияИСроком.НомерСтроки
    И МатериалыВЭксплуатацииОстаткиИОбороты.Номенклатура = ЭксплуатацияСНазначениемИспользованияИСроком.Номенклатура
    И МатериалыВЭксплуатацииОстаткиИОбороты.СерияНоменклатуры = ЭксплуатацияСНазначениемИспользованияИСроком.СерияНоменклатуры
    И МатериалыВЭксплуатацииОстаткиИОбороты.ХарактеристикаНоменклатуры = ЭксплуатацияСНазначениемИспользованияИСроком.ХарактеристикаНоменклатуры
    И МатериалыВЭксплуатацииОстаткиИОбороты.ФизЛицо = ЭксплуатацияСНазначениемИспользованияИСроком.ФизЛицо
    ПО ОстаткиМатериаловВЭксплуатации.СотрудникФизлицо = МатериалыВЭксплуатацииОстаткиИОбороты.ФизЛицо
    И ОстаткиМатериаловВЭксплуатации.Номенклатура = МатериалыВЭксплуатацииОстаткиИОбороты.Номенклатура
    И ОстаткиМатериаловВЭксплуатации.ХарактеристикаНоменклатуры = МатериалыВЭксплуатацииОстаткиИОбороты.ХарактеристикаНоменклатуры
    И ОстаткиМатериаловВЭксплуатации.СерияНоменклатуры = МатериалыВЭксплуатацииОстаткиИОбороты.СерияНоменклатуры
    ГДЕ
    МатериалыВЭксплуатацииОстаткиИОбороты.КоличествоПриход > 0
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо,
    ОстаткиНазначениеИДатаСдачи.Должность,
    ОстаткиНазначениеИДатаСдачи.Номенклатура КАК ФактическиВыданнаяСпецовка,
    ОстаткиНазначениеИДатаСдачи.ХарактеристикаНоменклатуры,
    ОстаткиНазначениеИДатаСдачи.НазначениеИспользования,
    ОстаткиНазначениеИДатаСдачи.СрокПолезногоИспользования,
    МАКСИМУМ(ОстаткиНазначениеИДатаСдачи.Дата) КАК Дата,
    МАКСИМУМ(ОстаткиНазначениеИДатаСдачи.ОриентировочнаяДатаСдачи) КАК ОриентировочнаяДатаСдачи,
    ПотребностиИРазмеры.РазмерОбуви,
    ПотребностиИРазмеры.РазмерОдежды,
    ПотребностиИРазмеры.Рост,
    ПотребностиИРазмеры.НазначениеИспользования КАК ДолжноеНазначениеИспользования,
    ПотребностиИРазмеры.Номенклатура КАК НеобходимаяСпецовка,
    ПотребностиИРазмеры.ФизЛицо
    ИЗ
    ПотребностиИРазмеры КАК ПотребностиИРазмеры
    ПОЛНОЕ СОЕДИНЕНИЕ ОстаткиНазначениеИДатаСдачи КАК ОстаткиНазначениеИДатаСдачи
    ПО ПотребностиИРазмеры.ФизЛицо = ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо
    И ПотребностиИРазмеры.Номенклатура = ОстаткиНазначениеИДатаСдачи.Номенклатура

    СГРУППИРОВАТЬ ПО
    ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо,
    ОстаткиНазначениеИДатаСдачи.Должность,
    ОстаткиНазначениеИДатаСдачи.Номенклатура,
    ОстаткиНазначениеИДатаСдачи.ХарактеристикаНоменклатуры,
    ОстаткиНазначениеИДатаСдачи.НазначениеИспользования,
    ОстаткиНазначениеИДатаСдачи.СрокПолезногоИспользования,
    ПотребностиИРазмеры.РазмерОбуви,
    ПотребностиИРазмеры.РазмерОдежды,
    ПотребностиИРазмеры.Рост,
    ПотребностиИРазмеры.НазначениеИспользования,
    ПотребностиИРазмеры.Номенклатура,
    ПотребностиИРазмеры.ФизЛицо
```

Проблема возникла в последней выборке из двух таблиц (рис 3). Соединяю по физлицу и номенклатуре (см снимок 3) Ожидаю результат как на снимке 1, но получаю другой (снимок 4), т.е. не соединил.
При этом, если соединяю только по физлицу, то соединяет (снимок 5, физлицо дважды - из обеих таблиц, потому что ищу ошибку).
Подскажите, где я косячу? Почему не соединяет по физлицу и номенклатуре одновременно?

Спасибо за ответы
---
я бы добавил поле

```1c
    ВЫБОР КОГДА ПотребностиИРазмеры.ФизЛицо ЕСТЬ NULL ТОГДА
    ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо
    ИНАЧЕ ПотребностиИРазмеры.ФизЛицо КАК ФИЗлицо
```

чтобы в результате не было двух полей ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо и ПотребностиИРазмеры.ФизЛицо и группировать по этому полю
---
Уверены на счет полного соединения?
---
Поскажите, какое использовать
---
видимо поменять местами таблицы и использовать левое.

Суть: есть несколько источников информации: выданная амуниция и пакет со сроком эксплуатации. Нужно получить список сотрудников с датой выдачи им чего-либо и датой следующей выдачи. Нужно определить, какая тут таблица является основной, а какая - зависимой. Дальше к основной таблице прилепить зависимую левым соединением. В принципе можно взять список работающих сотрудников, к сотрудникам прилепить пакеты левым соединением, к получившемуся результату этим же соединением прилепить предыдущую выдачу (последнюю выдачу - максимум по дате оборота или как-то так). Ну и все. 2 часа работы в принципе...
---
можно взять список работающих сотрудников, к сотрудникам прилепить пакеты левым соединением, - 100%
---
пакеты это потребность. А как сравнить с фактически выданным??
---
аналогично соединяете по физлицу...
---
я соединила по физлицу. С физлицом все ок, номенклатура перемешивается в кучку. Нужно соединение по физлицу и номенклатуре
---
датой выдачи им чего-либо и датой следующей выдачи. Нужно определить, какая тут таблица является основной, а какая - зависимой. Дальше к основной таблице прилепить зависимую левым соединением. В принципе можно взять список работающих сотрудников, к сотрудникам прилепить пакеты левым соединением, к получившемуся результату этим же соединением прилепить предыдущую выдачу (последнюю выдачу - максимум по дает оборота или как-то так). Ну и все.

Дело в том, что в тех условиях, которые я имею в данный момент, номенклатура в одной таблице и в другой не обязаны совпадать. Т.е. в пакете положенной спецодежды может быть написано "Ботинки сварщика", а фактически выданная номенклатура - "Ботинки ю/к" (потому что так купили и выдали). Если я возьму за основу Таблицу с тем, что человеку положено и сделаю левое соединение с тем, что по факту, то фактически выданные ботинки ю/к я потеряю. Мне нужна номенклатура из обеих таблиц, поэтому я и ухватилась за полное, ожидая, что в тех местах, где не совпадает, получу пустые значения.
---
в тех местах, где не совпадает, получу пустые значенияТут сложнее случай, но тоже решается. Я так понимаю, что сотруднику положен список амуниции, но выдают ему эту амуницию не всегда согласно списку. Тогда надо список сотрудников, к нему левым соединением выданную амуницию, а потом уже соединением положенную амуницию можно и полным прибить.
---
Вот. Я именно так и делаю. Запрос формирует кучку временных таблиц, на последнем этапе получается две таблицы: (рис 1)
Потом полным соединением по ФизЛицу и номенклатуре я пытаюсь получить таблицу (рис 2) и не получаю :(
---
покажите эту часть запроса, где идет соединение двух таблиц
---
выше есть, дублирую:

```1c
    ////////////////////////////////////////////////////////////////////////////////
    ВЫБРАТЬ
    ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо,
    ОстаткиНазначениеИДатаСдачи.Должность,
    ОстаткиНазначениеИДатаСдачи.Номенклатура КАК ФактическиВыданнаяСпецовка,
    ОстаткиНазначениеИДатаСдачи.ХарактеристикаНоменклатуры,
    ОстаткиНазначениеИДатаСдачи.НазначениеИспользования,
    ОстаткиНазначениеИДатаСдачи.СрокПолезногоИспользования,
    МАКСИМУМ(ОстаткиНазначениеИДатаСдачи.Дата) КАК Дата,
    МАКСИМУМ(ОстаткиНазначениеИДатаСдачи.ОриентировочнаяДатаСдачи) КАК ОриентировочнаяДатаСдачи,
    ПотребностиИРазмеры.РазмерОбуви,
    ПотребностиИРазмеры.РазмерОдежды,
    ПотребностиИРазмеры.Рост,
    ПотребностиИРазмеры.НазначениеИспользования КАК ДолжноеНазначениеИспользования,
    ПотребностиИРазмеры.Номенклатура КАК НеобходимаяСпецовка,
    ПотребностиИРазмеры.ФизЛицо
    ИЗ
    ПотребностиИРазмеры КАК ПотребностиИРазмеры
    ПОЛНОЕ СОЕДИНЕНИЕ ОстаткиНазначениеИДатаСдачи КАК ОстаткиНазначениеИДатаСдачи
    ПО ПотребностиИРазмеры.ФизЛицо = ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо
    И ПотребностиИРазмеры.Номенклатура = ОстаткиНазначениеИДатаСдачи.Номенклатура

    СГРУППИРОВАТЬ ПО
```
---
я бы добавил поле

```1c
    ВЫБОР КОГДА ПотребностиИРазмеры.ФизЛицо ЕСТЬ NULL ТОГДА
    ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо
    ИНАЧЕ ПотребностиИРазмеры.ФизЛицо КАК ФИЗлицо
```

чтобы в результате не было двух полей ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо и ПотребностиИРазмеры.ФизЛицо и группировать по этому полю
---
1 ПотребностиИРазмеры.ФизЛицо никогда не NULL
2 разве можно так сделать с полем, по которому соединяешь? Соединение же сопоставляет и ищет одинаковые
---
ПотребностиИРазмеры.ФизЛицо никогда не NULL - в теории при полном соединении там может быть NULL. Но если в таблице ПотребностиИРазмеры точно есть все физ лица, тогда можно оставить одно поле ПотребностиИРазмеры.ФизЛицо и группировать по нему
---
А, ну да, если физ лицу когда-либо выдана спецуха, но при этом по какой-то причине не назначен пакет положенной спецухи, то нулл вполне может получиться, соглашусь.
Не уверена, что проблема кроется здесь, но поиграюсь
---
разве можно так сделать с полем, по которому соединяешь? - можно, потому что соединение идет по полям ОстаткиНазначениеИДатаСдачи.СотрудникФизлицо и ПотребностиИРазмеры.ФизЛицо, а это дополнительное поле
---
Не уверена, что проблема кроется здесь, но поиграюсь
Вы были правы, я нет.
Действительно, слияние двух полей в одно решило данную проблему. Спасибо.
Глазом, правда, увидела еще нюанс, но это уже будет другая история (если сама не решу)